

A Different Approach to Create Swimmer Plot Using Proc Template and SGRENDER

Jui-Fu Huang, Baxalta, Cambridge, MA

ABSTRACT

Creating figures for statistical analysis of clinical trials are challenging when limited to specific SAS version and internal SOP. Previous work (Stacey Phillips, 2014; Sanjay Matange) has been done mainly with PROC SGPLOT. In this paper, I introduce a different way to create swimmer plot by using PROC TEMPLATE and PROC SGRENDER, to give additional information such as dose level, cancer type, or genetic type. Additionally, instead of marking "ongoing response" with annotation, I assign a subject-level format with PROC TEMPLATE function "BARLABELFORMAT" to simplify and automatically update output figures when receiving new data. This paper will demonstrate detailed steps of creating swimmer plot as well as compare the advantages and disadvantages of different methods.

INTRODUCTION

At PharmaSUG 2014, Stacey D. Phillips created swimmer plot with SG annotation and SGPLOT, with the need to present related tumor diagnosis information. This paper uses PROC TEMPLATE to add columns to the left of plot.

Furthermore, while annotation is a powerful yet flexible facility to add text or symbol into graphics, it can also be time consuming when we need to adjust its position, size, and font. To avoid using annotation, BARLAELFORMAT is a function that can easily achieve this goal.

METHODS

The example output demonstrated in this paper contains five categories of information:

Information	Variable Name
Treatment Group	ARM
Treatment Duration	TRT_DUR
Dose	DOSE
Genetic Information	GI
Treatment Status	STATUS

Table 1. Contents of Sample Data Set

	GI	STATUS	ARM	DOSE	TRT_DUR	SUBJID	_TRTDUR	START	END	LABEL
1	WT		ARM 2(QW)	10	9.1000020002	020002	9.1000020002	9.1000020002	9.1000020002	9.1
2	WT		ARM 1(Q2W)	10	8.0000010007	010007	8.0000010007	8.0000010007	8.0000010007	8.0
3	WT		ARM 2(QW)	10	7.7000020001	020001	7.7000020001	7.7000020001	7.7000020001	7.7
4	MUT		ARM 2(QW)	10	6.2000020007	020007	6.2000020007	6.2000020007	6.2000020007	6.2
5	WT	ONGOING	ARM 1(Q2W)	25	5.5000010012	010012	5.5000010012	5.5000010012	5.5000010012	5.5+
6	MUT		ARM 1(Q2W)	10	5.3000010008	010008	5.3000010008	5.3000010008	5.3000010008	5.3
7	MUT		ARM 2(QW)	10	4.2000020011	020011	4.2000020011	4.2000020011	4.2000020011	4.2
8	MUT		ARM 1(Q2W)	3	4.1000010004	010004	4.1000010004	4.1000010004	4.1000010004	4.1
9	MUT		ARM 1(Q2W)	25	4.0000010010	010010	4.0000010010	4.0000010010	4.0000010010	4.0
10	MUT		ARM 2(QW)	10	3.9000020010	020010	3.9000020010	3.9000020010	3.9000020010	3.9
11	WT		ARM 1(Q2W)	1	3.9000010001	010001	3.9000010001	3.9000010001	3.9000010001	3.9
12	WT	ONGOING	ARM 2(QW)	10	3.1000020004	020004	3.1000020004	3.1000020004	3.1000020004	3.1+
13	WT		ARM 1(Q2W)	3	3.1000010005	010005	3.1000010005	3.1000010005	3.1000010005	3.1
14	MUT		ARM 2(QW)	10	3.0000020009	020009	3.0000020009	3.0000020009	3.0000020009	3.0
15	MUT		ARM 1(Q2W)	10	2.9000010009	010009	2.9000010009	2.9000010009	2.9000010009	2.9
16	WT		ARM 1(Q2W)	3	2.6000010006	010006	2.6000010006	2.6000010006	2.6000010006	2.6
17	MUT		ARM 2(QW)	10	2.4000020008	020008	2.4000020008	2.4000020008	2.4000020008	2.4
18	WT		ARM 2(QW)	10	2.3000020003	020003	2.3000020003	2.3000020003	2.3000020003	2.3
19	MUT		ARM 2(QW)	10	2.1000020012	020012	2.1000020012	2.1000020012	2.1000020012	2.1
20	MUT		ARM 1(Q2W)	1	1.9000010003	010003	1.9000010003	1.9000010003	1.9000010003	1.9

Figure 1. Sample Data Set

First, a simple swimmer plot looked just like a sorted bar chart, you can find many examples creating such plot with BARCHART function of PROC TEMPLATE, see figure 2.

```

proc template;
  define statgraph F1;
    begingraph/ designwidth=600px designheight=410px;
    layout lattice/columns=1 columngutter=0;
    /*Swimmer Plot*/
    layout overlay/yaxisopts=(reverse=true display=none
      tickvalueattrs=(weight=bold))
      xaxisopts=(griddisplay=on labelattrs=(weight=bold)
        label="Treatment Duration (months)"
        linearopts=(tickvalueformat=(extractscale=true)));
    barchart x=SUBJID y=TRT_DUR/name="ARM" barwidth=0.6 orient=horizontal
      group=ARM grouporder=ascending
      outlineattrs=(color=black);

    discretelegend "ARM";
  endlayout;
endlayout;
endgraph;
end;
run;

proc sgrender data=F2 template=F1;
run;

```




Figure 2. A Simple Swimmer Plot

Next, to add some subject-level information, two SCATTERPLOT statements are used to show marker character at the left of plot, remember to adjust column weights to avoid plot being squeezed (figure 3).

```

proc template;
  define statgraph F1;
    begingraph/ designwidth=600px designheight=410px;
    layout lattice/columns=3 columngutter=2
      columnweights=(.12 .18 .70);
    column2headers;
    entry halign=center "Dose (mg/kg)"/textattrs=(size=7.5pt weight=bold);

```

```

    entry halign=center "Genetic Info"/textattrs=(size=7.5pt weight=bold);
endcolumn2headers;
/*Dose*/
layout overlay/walldisplay=none border=false
    yaxisopts=(display=none reverse=true type=discrete)
    xaxisopts=(display=none offsetmin=0.3 offsetmax=0);
    scatterplot y=SUBJID x=constant/markercharacter=DOSE markerattrs=(size=0);
endlayout;
/*Genetic Infomation*/
layout overlay/walldisplay=none border=false
    yaxisopts=(display=none reverse=true type=discrete)
    xaxisopts=(display=none offsetmin=0.3 offsetmax=0);
    scatterplot y=SUBJID x=constant/markercharacter=GI markerattrs=(size=0);
endlayout;
/*Swimmer Plot*/
layout overlay/yaxisopts=(reverse=true display=none
    tickvalueattrs=(weight=bold))
    xaxisopts=(griddisplay=on labelattrs=(weight=bold)
    label="Treatment Duration (months)"
    linearopts=(tickvalueformat=(extractscale=true)));
    barchart x=SUBJID y=TRT_DUR/name="ARM" barwidth=0.6 orient=horizontal
    group=ARM grouporder=ascending
    outlineattrs=(color=black);

    discretelegend "ARM";
endlayout;
endlayout;
endgraph;
end;
run;

proc sgrender data=F2 template=F1;
run;

```




Figure 3. Swimmer Plot with Dosage and Genetic Information

The last step is to put treatment status at the end of each bar. In this case, we want to know how long has each subject taken study treatment, and whether the subject is still taking it or had dropped from study. This can be done by adding a bar label, however, to prevent from two or more subjects with exactly the same treatment duration, which can mess up the bar label, each record requires a unique and none-overlapping format value. Here I concatenate treatment duration and subject id (which is unique! Or you can generate some non-repeat random number), and to avoid affecting bar length, I add some 0s right behind original treatment duration.

```

data F2;
  set F2;
  length _TRTDUR $15.;
  _TRTDUR=strip(put(TRT_DUR, 3.1))||"000"||strip(SUBJID);
  TRT_DUR=input(strip(_TRTDUR), 12.10);
  constant=1;
  START=TRT_DUR;
  if TRT_DUR>10 then END=TRT_DUR+0.00001;
  else END=TRT_DUR;
  if STATUS="ONGOING" then LABEL=put(START,4.1)||"+";
  else LABEL=put(START,4.1);
  retain FMTNAME "ONGOING" TYPE "n";
run;

data CTRL(keep=START END LABEL FMTNAME TYPE);
  set F2;
run;

proc format library=work cntlin=CTRL;
run;

proc template;
  define statgraph F2;
    begingraph/ designwidth=600px designheight=410px;
    layout lattice/columns=3 columngutter=2
      columnweights=(.12 .18 .70);
    column2headers;
      entry halign=center "Dose (mg/kg)"/textattrs=(size=7.5pt weight=bold);
      entry halign=center "Genetic Info"/textattrs=(size=7.5pt weight=bold);
    endcolumn2headers;
    /*Dose*/
    layout overlay/walldisplay=none border=false
      yaxisopts=(display=none reverse=true type=discrete)
      xaxisopts=(display=none offsetmin=0.3 offsetmax=0);
      scatterplot y=SUBJID x=constant/markercharacter=DOSE markerattrs=(size=0);
    endlayout;
    /*Genetic Infomation*/
    layout overlay/walldisplay=none border=false
      yaxisopts=(display=none reverse=true type=discrete)
      xaxisopts=(display=none offsetmin=0.3 offsetmax=0);
      scatterplot y=SUBJID x=constant/markercharacter=GI markerattrs=(size=0);
    endlayout;
    /*Swimmer Plot*/
    layout overlay/yaxisopts=(reverse=true label="Treatment Duration (months)"
      tickvalueattrs=(weight=bold)) display=none
      xaxisopts=(griddisplay=on labelattrs=(weight=bold)
      linearopts=(tickvalueformat=(extractscale=true)));
      barchart x=SUBJID y=TRT_DUR/name="ARM" barlabel=true barlabelformat=ONGOING.
        barwidth=0.6 orient=horizontal
        group=ARM grouporder=ascending outlineattrs=(color=black);
      discretelegend "ARM";
    endlayout;
  endlayout;
endgraph;
end;
run;

```

```
proc sgrender data=F2 template=F2;
run;
```

Now, by adding BARLABEL, we have a final version of swimmer plot with all the requested information. Notice that numbers with a "+" sign means that treatment is still ongoing for this subject.




Figure 4. Swimmer Plot with Treatment Status

CONCLUSION

Creating ad hoc outputs based on the requirements of analysis and version/package of SAS always forces you to use different combination of procedures and functions. While annotation certainly has more flexibility, choosing a ready-to-use function without inputting extra settings manually can save you some efforts, and shorten the time you needed to update your program whenever new data comes in.

REFERENCES

Phillips, Stacey (2014) "Swimmer Plot: Tell a Graphical Story of Your Time to Response Data Using PROC SGPLOT" *Proceedings of the Pharmasug 2012 Conference*. Cary, NC: SAS Institute.

Matange, Sanjay. "Swimmer plot." Cary, NC: SAS Institute.
<http://blogs.sas.com/content/graphicallyspeaking/2014/06/22/swimmer-plot/>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Jui-Fu Huang
 Enterprise: Baxalta
 Address: 650 East Kendall St. Floor 4
 City, State ZIP: Cambridge, MA 02142
 Work Phone: (617) 588-8257
 E-mail: jui.fu.huang@baxalta.com
 Web: www.baxalta.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.